
//********************************************************************************************
//*   MACRO: HEADFOOT.WCM
//*   PURPOSE: Add Headers/Footers to your Presentations Slide Show.
//********************************************************************************************
APPLICATION (Presentations; "Presentations"; Default; "EN")

PWidthW:=EnvFormInfo(1)
PHeightW:=EnvFormInfo(0)
vMLeft:=EnvFormInfo(4)
vMTop:=EnvFormInfo(2)
vMRight:=EnvFormInfo(5)
vMBottom:=EnvFormInfo(3)

vXLeft:=(vMLeft)
vXRight:=(PWidthW - vMRight)

// Header
vXBottom:=(PHeightW - (vMTop+ 150w))
vXTop:=(vXBottom - 600w)

// Footer
vXXBottom:=(vMBottom + 400w)
vXXTop:=(vXXBottom - 600w)

GLOBAL vSlideNum:=1
GLOBAL vSlideNum2:=1
GLOBAL vMode:=0
GLOBAL vApply:=0 // 0-Apply All  1-Apply to Current
GLOBAL vStepNum:=0
GLOBAL vStepStay:=0
Call(CheckPRState)
vHeader:=1
vFooter:=0
vLeft:="Date"
vCenter:="Slide title"
vRight:="Slide number"
vTextColor:="Blue"
GLOBAL RGBColors[]={0;0;0}
vMin:=0
vDel:=0

Call(Pick)
Call(MainDialog)

LABEL(Pick)
DIALOGDEFINE("Pick"; 50; 50; 105; 95; Percent!|OK!; "Headers/Footers")
DIALOGADDGROUPBOX("Pick"; 101; 12; 6; 80; 54; "Select")
DIALOGADDRADIOBUTTON("Pick"; 102; 28; 22; 40; 10; "&Header"; vHeader)
DIALOGADDRADIOBUTTON("Pick"; 103; 28; 38; 40; 10; "&Footer"; vFooter)
DIALOGSHOW("Pick"; "Presentations")
DIALOGDESTROY("Pick")

RETURN

LABEL(MainDialog)
IF(vHeader = 1)

DIALOGDEFINE(100;50;50;310;130;Percent!|Cancel!;"Create Header")
ELSE

DIALOGDEFINE(100;50;50;310;130;Percent!|Cancel!;"Create Footer")
ENDIF
DIALOGADDGROUPBOX(100; 101; 4; 3; 86; 32; "&Left")
DIALOGADDGROUPBOX(100; 102; 110; 3; 86; 32; "&Center")
DIALOGADDGROUPBOX(100; 103; 218; 3; 86; 32; "&Right")
DIALOGADDVLINE(100; 104; 99; 6; 66)



DIALOGADDVLINE(100; 105; 207; 6; 66)
DIALOGADDCOMBOBOX(100; 106; 12; 15; 72; 70; Sorted!|Droplist!; vLeft; 0)
DIALOGADDCOMBOBOX(100; 107; 118; 15; 72; 70; Sorted!|Droplist!; vCenter; 0)
DIALOGADDCOMBOBOX(100; 108; 226; 15; 72; 70; Sorted!|Droplist!; vRight; 0)
FOR(x; 106; x<109; x+1)

DialogAddListItem(100; x; "Date")
DialogAddListItem(100; x; "User input")
DialogAddListItem(100; x; "Slide number")
DialogAddListItem(100; x; "Blank")
DialogAddListItem(100; x; "Filename")
DialogAddListItem(100; x; "Slide title")

ENDFOR
DIALOGADDTEXT(100; 109; 3; 38; 35; 11; 1; "User input:")
DIALOGADDTEXT(100; 110; 114; 38; 35; 11; 1; "User input:")
DIALOGADDTEXT(100; 111; 222; 38; 35; 11; 1; "User input:")
DIALOGADDEDITBOX(100; 112; 8; 49; 70; 11; Left!; vLeftText; 125)
DIALOGADDEDITBOX(100; 113; 116; 49; 70; 11; Left!; vCenterText; 125)
DIALOGADDEDITBOX(100; 114; 224; 49; 70; 11; Left!; vRightText; 125)
DIALOGADDHLINE(100; 115; 3; 70; 305)
DIALOGADDTEXT(100; 116; 8; 77; 40; 11; 1; "&Text color:")
DIALOGADDCOMBOBOX(100; 117; 13; 90; 72; 70; Sorted!|Droplist!; vTextColor; 0)

DialogAddListItem(100; 117; "Red")
DialogAddListItem(100; 117; "White")
DialogAddListItem(100; 117; "Black")
DialogAddListItem(100; 117; "Blue")
DialogAddListItem(100; 117; "Green")
DialogAddListItem(100; 117; "Yellow")
DialogAddListItem(100; 117; "Gray (Light)")
DialogAddListItem(100; 117; "Gray (Dark)")
DialogAddListItem(100; 117; "Magenta")
DialogAddListItem(100; 117; "Cyan")

DIALOGADDCHECKBOX(100; 119; 155; 81; 160; 10; "Minimi&ze Presentations for faster changes"; vMin)
DIALOGADDCHECKBOX(100; 121; 155; 93; 125; 10; "&Delete previous headers/footers."; vDel)
DIALOGADDPUSHBUTTON(100; "APPLYButton"; 166; 111; 40; 13; NonDefaultBttn!; " &Apply ")
DIALOGADDPUSHBUTTON(100; "APPLYALLButton"; 212; 111; 45; 13; NonDefaultBttn!; " A&pply to All ")
DIALOGLOAD(100; "Presentations")
RegionEnableWindow("100.109"; Disable!)
RegionEnableWindow("100.110"; Disable!)
RegionEnableWindow("100.111"; Disable!)
RegionEnableWindow("100.112"; Disable!)
RegionEnableWindow("100.113"; Disable!)
RegionEnableWindow("100.114"; Disable!)
IF(vMode = 1)

RegionEnableWindow("100.APPLYALLButton"; Disable!)
ENDIF
RegionSetFocus("100.CancelBttn")
DIALOGSHOW(100;;MainLoop)
CallBackWait()
DialogDestroy(100)
Call(IsSaved)
Call(NumberofSlide)
IF(vMin = 1)

PRActivate ()
ApplicationMinimize ()

ENDIF
WaitMsgInit ()
WaitMsgDisplay ("Adding Header/Footer... "; 0)
Call(AddHeaderFooter)
WaitMsgHide ()
WaitMsgDestroy ()
IF(vMin = 1)

ApplicationRestore ()



ENDIF
QUIT

LABEL(MainLoop)
if(MainLoop[5]=274)

DialogDestroy(100)
QUIT

endif
if(MainLoop[5]=273)

switch(MainLoop[3])
caseof"APPLYButton":

DialogDismiss("100";"APPLYButton")
vApply:=1
CallBackResume()

caseof"APPLYALLButton":
DialogDismiss("100";"APPLYALLButton")
vApply:=0
CallBackResume()

caseof"CancelBttn":
DialogDestroy(100)
QUIT

caseof"106":
vLeft = RegionGetSelectedText("100.106")
IF(vLeft = "User input")

RegionEnableWindow("100.109"; Enable!)
RegionEnableWindow("100.112"; Enable!)

ELSE
RegionEnableWindow("100.109"; Disable!)
RegionEnableWindow("100.112"; Disable!)

ENDIF
caseof"107":

vCenter = RegionGetSelectedText("100.107")
IF(vCenter = "User input")

RegionEnableWindow("100.110"; Enable!)
RegionEnableWindow("100.113"; Enable!)

ELSE
RegionEnableWindow("100.110"; Disable!)
RegionEnableWindow("100.113"; Disable!)

ENDIF
caseof"108":

vRight = RegionGetSelectedText("100.108")
IF(vRight = "User input")

RegionEnableWindow("100.111"; Enable!)
RegionEnableWindow("100.114"; Enable!)

ELSE
RegionEnableWindow("100.111"; Disable!)
RegionEnableWindow("100.114"; Disable!)

ENDIF
endswitch

endif
RETURN

LABEL(ErrorX)
// Just return back.

RETURN

LABEL(AddHeaderFooter)
IF((vMode = 2)AND(vApply = 0))

SlideShowSlide()
ShowSlide(1)

ENDIF
UnselectAllObjects()



FOR(x; 1; x<(vSlideNum+1); x+1)
WaitMsgDisplay ("Adding Header/Footer... "; vStepNum)
vStepNum:=(vStepNum + vStepStay)

IF(vDel = 1)
OnError Call(ErrorX)
UnselectAllObjects ()
BeginSelect ()
ObjectPointSelect (vMLeft;  (vXTop + 600w); No!)
vHeight:=EnvGetObjectHeight()
vWidth:=EnvGetObjectWidth()
IF((vHeight < 265W)AND(vHeight > 240W))

DeleteObjects ()
ENDIF
UnselectAllObjects ()
BeginSelect ()
ObjectPointSelect (vMLeft; (vXXTop + 600w); No!)
vHeight:=EnvGetObjectHeight()
vWidth:=EnvGetObjectWidth()
IF((vHeight < 265W)AND(vHeight > 240W))

DeleteObjects ()
ENDIF

ENDIF
IF(vHeader = 1)

AddTextBox (vXLeft; vXBottom; vXRight; vXTop)
ELSE

AddTextBox (vXLeft; vXXBottom; vXRight; vXXTop)
ENDIF
ObjectSetFrameState (FrameState: FrameOff!)
ObjectSetFillState (FillState: Filled!)
SetFontFace (FontName: "Arial")
SetFontFace (PointSize: 12)
RGBColors[]=ConvertColor(vTextColor; RGBColors[])
SetFontPattern (BrushPattern: Solid!; Grad1Red:RGBColors[1]; Grad1Green:RGBColors[2]; Grad1Blue:RGBColors[3]; 

Grad2Red: 255; Grad2Green: 255; Grad2Blue: 255; 0; 0; 0; PenFRed: RGBColors[1]; PenFGreen:RGBColors[2]; 
PenFBlue:RGBColors[3]; PenHorzWidth: 0; PenVertWidth: 0; eFlags: FramedClosedFilled!; PenStyle: 0; PenPattern: 0)

RealizeFontChanges ()
TextTab ()
IF(vLeft != "User input")

DoInput(vLeft; x)
ELSE

Type(Text: vLeftText)
ENDIF
TextLayoutCenter ()
IF(vCenter != "User input")

DoInput(vCenter; x)
ELSE

Type(Text: vCenterText)
ENDIF
TextLayoutFlushRight ()
IF(vRight != "User input")

DoInput(vRight; x)
ELSE

Type(Text: vRightText)
ENDIF
Type (Text: "     ")
TextEditExit ()
UnselectAllObjects ()
IF((vMode = 2)AND(vApply = 0))

SlideNext()
ENDIF

ENDFOR
WaitMsgDisplay ("Adding Header/Footer... "; 100)



RETURN

PROCEDURE DoInput(vInput; vNum)
IF(vInput = "Date")

DateText ()
ENDIF
IF(vInput = "Slide number")

IF(vMode = 2)
IF(vApply = 0)

Type(Text: "Slide "+vNum+" of "+vSlideNum)
ELSE

vNumX:=EnvCurrentSlideNumber()
Type(Text: "Slide "+vNumX+" of "+vSlideNum2)

ENDIF
ENDIF
IF(vMode = 1)

Type(Text: "Page 1")
ENDIF

ENDIF
IF(vInput = "Blank")
ENDIF
IF(vInput = "Filename")

vSfilename:=EnvCurrentFilename
IF(vSfilename != "")

Type(Text: vSfilename)
ELSE

Type(Text: "(Unsaved)")
ENDIF

ENDIF
IF(vInput = "Slide title")

IF(vMode = 2)
vStitle:=EnvSlideTitle()
IF(vStitle != "")

Type(Text: vStitle)
ELSE

Type(Text: "(Untitled)")
ENDIF

ENDIF
IF(vMode = 1)

Type(Text: "(Untitled)")
ENDIF

ENDIF
ENDPROC

LABEL(NumberofSlide)
IF(vMode = 2)

vSlideNum2:=EnvNumberOfSlides()
ENDIF
IF((vMode = 2)AND(vApply = 0))

vSlideNum:=EnvNumberOfSlides()
ELSE

vSlideNum:=1
ENDIF
vStepStay:=(100 / vSlideNum)

RETURN

Label(CheckPRState)
vMode:=EnvCurrentMode()
if((vMode != 1)AND(vMode != 2))

BEEP
MESSAGEBOX(vStatus; "Header/Footer Error"; "Presentations must be in the Slide Editor or Drawing mode."; IconStop!)
Quit



endif
Return

LABEL(IsSaved)
IF((vLeft = "Filename")OR(vCenter = "Filename")OR(vRight = "Filename"))
vIsSaved:=EnvCurrentFilename

IF(vIsSaved = "")
MessageBox(vYesNo; "File not saved"; "You are requesting a ""filename"" as a Header/Footer, but you do not have your 

document saved.  A ""(Unsaved)"" will be inserted.  Would you like to continue?"; YesNo! | IconStop!)
IF(vYesNo = 7)  // NO

QUIT
ENDIF

ENDIF
ENDIF

RETURN

Function ConvertColor(vColor; vRGB[]) // 1=RED, 2=GREEN, 3=BLUE
vRGB[1]:=0
vRGB[2]:=0
vRGB[3]:=0
if(vColor = "Black")

// No change needed
endif
if(vColor = "White")

vRGB[1]:=255
vRGB[2]:=255
vRGB[3]:=255

endif
if(vColor = "Red")

vRGB[1]:=255
endif
if(vColor = "Blue")

vRGB[3]:=255
endif
if(vColor = "Green")

vRGB[2]:=255
endif
if(vColor = "Cyan")

vRGB[2]:=255
vRGB[3]:=255

endif
if(vColor = "Magenta")

vRGB[1]:=255
vRGB[3]:=255

endif
if(vColor = "Yellow")

vRGB[1]:=255
vRGB[2]:=255

endif
if(vColor = "Gray (Light)")

vRGB[1]:=191
vRGB[2]:=191
vRGB[3]:=191

endif
if(vColor = "Gray (Dark)")

vRGB[1]:=124
vRGB[2]:=124
vRGB[3]:=124

endif
Return(vRGB[])
EndFunc



//***************************************************************
// Wait Message Procedures
//***************************************************************
PROCEDURE WaitMsgInit ()

DialogDefine ("WaitMsg"; 50; 50; 150; 50; 16; "Please Wait")
DialogAddText ("WaitMsg"; "WaitText"; 8; 10; 144; 16; 1; "")
DialogAddProgress("WaitMsg"; "WaitBar"; 10; 30; 130; 15)
DialogLoad ("WaitMsg"; "Presentations")
Return

ENDPROC
PROCEDURE WaitMsgDisplay (Text; Number)

RegionSetWindowText ("WaitMsg" + ".WaitText"; Text)
RegionSetProgressPercent("WaitMsg" + ".WaitBar"; Number)
DialogShow ("WaitMsg";; WaitDlgCallBack)

ENDPROC
PROCEDURE WaitDlgCallBack ()

If (WaitDlgCallBack[5] = 274 And WaitDlgCallBack[6] = 61536)
Assert (CancelCondition!)

EndIf
ENDPROC
PROCEDURE WaitMsgHide ()

DialogDismiss ("WaitMsg"; "WaitText")
ENDPROC
PROCEDURE WaitMsgDestroy ()

DialogDestroy ("WaitMsg")
ENDPROC
//***************************************************************


